

Shannon Way Tewkesbury Gloucester GL20 8ND Ph: 011.44.1684.292333 Fx: 011.44.1684.297929 1000 Gamma Drive Suite 206 Pittsburgh, PA 15238 Ph: 412.968.9744 Fx: 412.968.9746 www.triomotion.com

# TECHNICAL NOTE

Doc No.: TN30-20

Date: December 19, 2005

Version: 1.0

Subject: ServoStar S300 SERCOS Registration

## **OVERVIEW:**

This document describes how to setup registration over a SERCOS ring using a Trio Motion Controller with the ServoStar S300 drive. It will include a TrioBASIC program that initializes all the necessary SERCOS parameters (IDNs) for a single axis. Even if you are not using registration, it may also be useful for getting started with using SERCOS on the S300. Once the ring is established, the standard TrioBASIC registration commands (REGIST, MARK, REG\_POS) will be available for the SERCOS axis.

With a traditional servo axis the registration signal is wired directly to an input on the Trio MC. The axis position typically comes from an encoder also wired directly to the Trio MC. When this input signal is asserted the Trio MC captures the position of the axis. This position capture is performed directly in hardware at a very high speed (typically  $1 - 2 \mu sec$ ).

In a SERCOS system the traditional +/- 10 volt command and the encoder feedback signal wiring are replaced by a digital fiber optic network. Since the axis encoder signal is not wired back to the Trio MC hardware, the high speed position capture is best performed inside the S300. This is triggered by wiring the registration signal to one of the inputs on the S300. The Trio MC simply arms the registration function in the S300 and receives the captured position via SERCOS.

#### REFERENCES:

- 1. Trio Application Note TN20-53, SERCOS Interface Commands from TrioBASIC
- 2. Trio Application Note TN20-63, SERCOS Registration and Homing
- 3. Kollmorgen SERCOS IDN Reference Guide for digital servoamplifiers

## **EQUIPMENT:**

- 1. Trio Motion Controller (firmware v1.64 or higher) with P291 SERCOS daughter board installed
- 2. Trio MotionPerfect2 software (v2.2.2.2 or higher) and communications cable
- 3. Kollmorgen ServoStar S300 drive (firmware v1.00 or higher) with SERCOS option card installed
- 4. Kollmorgen ServoStar S300 setup software (DriveGUI.exe) and communications cable
- 5. SERCOS cables, motor and cables

## **SETUP:**

- 1. Use the S300 GUI to setup the drive for the motor being used. This example uses SERCOS Position Mode, so the velocity and position loop tuning parameters must be set in the S300.
- 2. In the S300 GUI the SERCOS Baudrate and Address should be configured as follows:




Fig. 1 - S300 GUI SERCOS configuration screen.

- 3. Wire the 24 VDC registration signal to the X3 DIGITAL-IN1 on the S300.
- 4. Connect the SERCOS fiber optic transmit and receive cables between the S300 and the Trio MC.
- 5. Use MotionPerfect2 to load the *S300\_REG.BAS*, *STARTUP.BAS*, and *REG\_TEST.BAS* TrioBASIC program files (see appendix for program listings) into the Trio MC.

## **OPERATION:**

Once the above setup is complete we can proceed with controlling the S300 via SERCOS and testing the registration function.

- 1. Make sure the main bus power and logic power are applied to the S300. The S300 ENABLE signal can also be hardwired to the +24VDC logic power. The S300 display should indicate "P" followed by a number indicating the S300 current rating.
- 2. In MotionPerfect open the Terminal window #5 to see program debug messages. Run the \$300\_REG.BAS program to bring up the SERCOS ring. Then run the \$500\_REG.BAS program to set the Trio axis parameters and enable to drive. The \$500 display should now indicate an "E" followed by a number indicating the current rating, and there should be holding torque on the motor shaft. At this point, the axis is ready to perform all the standard TrioBASIC motion commands.

The Controller Configuration window should indicate that AXIS(1) is "SERCOS Position + Regist" axis type:




Fig. 2 - MotionPerfect Controller Configuration window.

3. Run the *REG\_TEST.BAS* program and the motor should slowly move forward. Provide the 24VDC registration signal to the S300 DIGITAL-IN1 terminal, then the captured registration position (REG\_POS) and current measured position (MPOS) are printed in the Terminal #5 window in MotionPerfect.

In this example the Trio AXIS(1) is being used, which happens to be the same as the S300 SERCOS drive address.

The number (n) used in the REGIST(n) command is arbitrary because the registration operation is defined by setting the IDNs in the S300.

The S300\_REG.BAS program can be modified to handle multiple SERCOS axes. In a multiple axis SERCOS configuration all axes must be stepped through the CP phases together. Some of the IDNs must be incremented to unique values for each of the axes. A multiple axis SERCOS initialization example program for the S300 is also available. If you are new to SERCOS it is recommended to get a single axis running properly first.

## **APPENDIX:**

```
' Filename: REG_S300.BAS
' Release: 19 Dec 2005
' Abstract: Simple TrioBASIC SERCOS setup for an S300 drive.
' Notes:
' 1) First tune S300 using the S300 GUI, and set SERCOS Address and Baudrate.
' 2) Open MotionPerfect2 Terminal window (see prt below) for debug messages.
' 3) Run REG_S300.BAS first, then STARTUP.BAS, then REG_TEST.BAS.
drv=1 'sercos drive id
ax=1 'trio axis number
prt=5 'motionperfect terminal for debug messages
slt=0 'daugther board slot
kollorgen=5 'vender specific comms
mode=1 '1=position, 0=velocity
reg=1 '1=config for registration, 0=no registration
TABLE(0, 124, 47, 45, 92, 124, 47, 45, 92) 'for pause display
'--Make sure Trio axis is available for SERCOS.
BASE(ax)
IF ATYPE<>0 THEN
  IF ATYPE<16 THEN
   PRINT #prt, "Axis(";ax[0];") ATYPE not available for SERCOS"
   STOP 'stop program on failure
  ENDIF
ENDIF
WHILE MTYPE<>0 : CANCEL : WEND 'cancel any commanded motions
WAIT IDLE 'wait until cancel is completed
SERVO=OFF : WDOG=OFF 'disable axis
DATUM(0) 'clear any motion errors
SERCOS_PHASE SLOT(slt)=0
PRINT #prt, "CP0" 'communication phase 0
```

```
SERCOS(2,slt,3,4)'set intensity=3, 4mhz
SERCOS(3,slt,0,0)'reset all axis definitions
IF SERCOS(10,slt)<>1 THEN
 PRINT #prt, "SERCOS loop open or distorted"
 STOP 'stop program on failure
ENDIF
SERCOS PHASE SLOT(slt)=1
PRINT #prt, "CP1" 'communication phase 1
IF SERCOS(7,slt,drv)<>1 THEN
  PRINT #prt, "Drive not found"
  STOP 'stop program on failure
ENDIF
SERCOS(3,slt,drv,ax,kollmorgen) 'map drv number to trio ax number
'-----
SERCOS_PHASE SLOT(slt)=2
PRINT #prt, "CP2" 'communication phase 2
SERCOS(6,slt,drv,99) 'clear any drive faults
GOSUB pause
SERCOS(5,slt,drv,1,2,1000) 'ID1 NC Cycle Time
SERCOS(5,slt,drv,2,2,1000) 'ID2 SERCOS Cycle Time
SERCOS(5,slt,drv,6,2,50) 'ID6 AT Tx Start Time
SERCOS(5,slt,drv,7,2,700) 'ID7 fb acquisition start time
SERCOS(5,slt,drv,8,2,800) 'ID8 Command Valid Time
SERCOS(5,slt,drv,9,2,1) 'ID9 Initial Address in MDT
SERCOS(5,slt,drv,10,2,8) 'ID10 Length of MDT
SERCOS(5,slt,drv,89,2,500) 'ID89 MDT tx Start Time
IF mode=1 THEN
  '--Position Mode
  SERCOS(5,slt,drv,15,2,7)'Custom
  SERCOS(5,slt,drv,32,2,3)'Position
  SERCOS(5,slt,drv,16,6,51)'AT Pos FB
```

```
SERCOS(5,slt,drv,24,6,47)'MDT Pos Cmd
ELSE
  '--Velocity Mode
  SERCOS(5,slt,drv,15,2,7)'Custom
  SERCOS(5,slt,drv,32,2,2)'velocity
  SERCOS(5, slt, drv, 16, 6, 51) 'AT Pos FB
  SERCOS(5,slt,drv,24,6,36)'MDT Velocity Cmd
ENDIF
IF reg=1 THEN
  '--Registration Setup
  SERCOS(5,slt,drv,16,6,51,130)'AT Pos FB + reg
  SERCOS(5,slt,drv,169,2,1) 'reg probe control
  SERCOS(5,slt,drv,301,2,405) 'regist controls probe 1 enable
  SERCOS(5,slt,drv,305,2,409) 'mark
  SERCOS(5,slt,drv,32768+3000,2,26) 'IDNP3001 IN1MODE motor pos latch
  SERCOS(5,slt,drv,32768+3001,2,26) 'IDNP3001 IN2MODE motor pos latch
  SERCOS(5,slt,drv,32768+3018,2,0) 'motor pos latch
ELSE
  SERCOS(5,slt,drv,16,6,51)'AT Pos FB
  SERCOS(5,slt,drv,169,2,0) 'reg probe control
  SERCOS(5,slt,drv,301,2,0) 'regist controls probe 1 enable
  SERCOS(5,slt,drv,305,2,0) 'mark
ENDIF
'--Drive Modulus Setup
SERCOS(5,slt,drv,79,4,65536) 'IDN79 motor counts per rev
SERCOS(5,slt,drv,103,4,16777216) 'IDN103 position modulo value
IF SERCOS(6,slt,drv,127)<>1 THEN
  PRINT #prt, "Drive failed CP2-CP3 check"
  STOP 'stop program on failure
ENDIF
SERCOS_PHASE SLOT(slt)=3
PRINT #prt, "CP3" 'communication phase 3
GOSUB pause
```

```
IF SERCOS(6,slt,drv,128)<>1 THEN
  PRINT #prt, "Drive 1 failed CP3-CP4 check"
  STOP 'stop program on failure
ENDIF
SERCOS_PHASE SLOT(slt)=4
PRINT #prt, "CP4" 'communication phase 4
GOSUB pause
IF mode=1 THEN
  IF reg=1 THEN
    ATYPE=23 'sercos position w/registration
    SERCOS(5,slt,drv,170,2,3) 'reg probing on
  ELSE
    ATYPE=17 'sercos position
    SERCOS(5,slt,drv,170,2,0) 'reg probing off
  ENDIF
ELSE
  IF reg=1 THEN
    ATYPE=22 'sercos velocity w/registration
    SERCOS(5,slt,drv,170,2,3) 'reg probing on
  ELSE
    ATYPE=16 'sercos velocity
    SERCOS(5,slt,drv,170,2,0) 'reg probing off
  ENDIF
ENDIF
PRINT #prt, "SERCOS init successful"
PRINT #prt, "Make sure S300 ENABLE signal made before enabling axis"
STOP 'end of program
pause:
  PRINT #prt, "Please Wait"
  FOR ncount = 0 TO 5
    FOR nindex= 0 TO 7
      PRINT #prt,CHR(TABLE(nindex));CHR(13);
      WA(100)
```

```
NEXT nindex
 NEXT ncount
RETURN
'-----
·------
' Filename: STARTUP.BAS
' Release: 19 Dec 2005
' Abstract: Sets Trio axis parameters for S300 SERCOS axis.
' Notes:
' 1) No gain parameters are required for SERCOS position mode.
' 2) Motor resolution is 65536 cts/rev (see IDN79).
'-----
'Start Standard Section
' Axis Parameters
BASE(1)
P_GAIN=0.0
I_GAIN=0.0
D_GAIN=0.0
OV_GAIN=0.0
VFF_GAIN=0.0
UNITS=1.0
SPEED=65536.0
ACCEL=6553600.0
DECEL=6553600.0
CREEP=32768.0
JOGSPEED=32768.0
FE_LIMIT=16383.0
DAC=0
SERVO=0
REP DIST=65536.0
FWD_IN=-1
REV_IN=-1
DATUM_IN=-1
FHOLD_IN=-1
FS_LIMIT=19660800.0
```

RS\_LIMIT=-19660800.0

```
'Stop Standard Section
DATUM(0) 'clear any motion errors
SERVO=ON 'close servo loop
WDOG=ON 'enable drive
'-----
' Filename: REG TEST.BAS
' Release: 19 Dec 2005
' Abstract: Simple test for S300 registration with SERCOS. Operation of
         REGIST(), MARK, and REG_POS are tested.
' Notes:
' 1) Wire a 24 volt registration signal into the S300 DIGITAL-IN1.
' 2) Open Terminal window# <prt> for messages.
'-----
drv=1 'sercos drive id
ax=1 'trio axis number
prt=5 'motionperfect terminal for debug messages
slt=0 'daugther board slot
BASE(ax)
REP_DIST=SERCOS(4,slt,drv,79,4) 'set repeat distance to motor resolution
REP_OPTION=1 'set rollover mode (0 to rep_dist)
WHILE MTYPE<>0 : CANCEL : WEND 'cancel any existing motion commands
WAIT IDLE
DEFPOS(0) 'set position to zero
FORWARD 'start motion
WHILE 1=1 'main loop
 REGIST(3) 'arm reg, number is arbitrary
 PRINT #prt, "Waiting for reg mark..."
 WAIT UNTIL MARK 'wait for reg mark
 PRINT #prt,"REG_POS=";REG_POS[0], "MPOS=";MPOS[0]
WEND
'------
```